

# Mathematica troubleshooting

## Math 3350

### Prof. Kevin Long

#### First rule of computer work: **DETAILS MATTER**

Computers are literal-minded, and do **exactly** what you **tell** them to do. It's critical to use very precise language.

In conversation with a human, I can write  $\cos(x)$ ,  $\text{Cos}(x)$ ,  $\cos[x]$ ,  $\text{Cos}[x]$ ,  $\cos x$ , or  $\text{Cos } x$ , and the human will understand from context that I mean "cosine of  $x$ ." A computer cannot understand anything from context. If the cosine function is defined to be  $\text{Cos}[x]$ , a computer cannot understand that  $\cos[x]$  is supposed to be a cosine, no matter how obvious it is to **you** that it's a cosine.

If, in an example, I have typed  $\text{Integrate}[\text{Cos}[x], x]$ , you will only get the same results if you use **exactly** the same syntax, including spelling, capitalization, commas, brackets `[]` instead of parentheses `()`, and so on.

#### Capitalize built-in function names, and use brackets `[]` not parentheses `()` for function arguments

It's  $\text{Cos}[x]$ , not  $\cos[x]$ ,  $\text{Cos}(x)$ ,  $\cos(x)$ , or  $\cos x$ .

$\cos[\text{Pi}]$  doesn't do anything, because the  $\cos$  function is undefined.

```
cos[Pi]
cos( $\pi$ )
```

Try  $\text{Cos}[\text{Pi}]$  instead, which will evaluate  $\cos \pi = -1$  correctly:

```
Cos[Pi]
-1
```

Use either  $\text{Exp}[x]$  or  $E^x$  for the exponential function  $e^x$ , not  $e^x$ .

```
f[x_] = e^x
e^x
```

Here are some examples of what will happen: derivatives and integrals won't be simplified correctly.

```
Integrate[f[x], x]
 $\frac{e^x}{\log(e)}$ 
D[f[x], x]
e^x log(e)
```

These calculations are actually correct, but they're not in the form you'd expect because *Mathematica* can't know that you intend the symbol "e" to mean the number  $e \approx 2.718281828459045$ . Because it doesn't know what is meant by "e", the best it can do is leave  $\log(e)$  in the expressions, unevaluated. The log of  $e$  is one, of course.

Instead, use  $E$  or the  $\text{Exp}[]$  function:

```
Integrate[Exp[x], x]
```

$$e^x$$

or

```
Integrate[E^x, x]
```

$$e^x$$

That's better.

## Similar errors with familiar functions

What's wrong with

```
Integrate[Cos[pi x], {x, 0, 1}]
```

$$\frac{\sin(\pi)}{\pi}$$

and

```
Integrate[cos[Pi x], {x, 0, 1}]
```

$$\int_0^1 \cos(\pi x) dx$$

and

```
Integrate[cos[pi x], {x, 0, 1}]
```

$$\int_0^1 \cos(\pi x) dx$$

and

```
Integrate[Cos Pi x, {x, 0, 1}]
```

$$\frac{\cos \pi}{2}$$

and

```
Integrate[CosPix, {x, 0, 1}]
```

CosPix


Can you do the calculation correctly?

## You can't plot a function with undefined symbols in it

If you see an empty plot, check for undefined symbols. For example,

```
ClearAll[a]
```

```
Plot[Exp[a x], {x, 0, 1}]
```




It's impossible to plot `Exp[a x]` because `a` hasn't been defined, and so the numerical values of  $e^{ax}$  can't be computed. If you give  $a$  a value, you can do the plot

```
a = Sqrt[2] / 2
```

$$\frac{1}{\sqrt{2}}$$


```
Plot[Exp[a x], {x, 0, 1}]
```



Be careful with operator precedence

$E^2 x$  and  $E^{(2 x)}$  are not the same, as you can see in the plot below

```
Plot[{E^2 x, E^ (2 x)}, {x, 0, 2}]
```



Do you see what caused the problem?

I usually use `Exp[ z ]` instead of `E^z`, in part to help avoid this problem.

## Trouble plotting derivatives

It's easy to differentiate

```
D[Sin[x], x]
cos(x)
```

so you'd think it would be easy to plot a derivative

```
Plot[D[Sin[x], x], {x, -1, 1}]
```


```
General::ivar: -0.999959 is not a valid variable. >>
```

```
General::ivar: -0.959143 is not a valid variable. >>
```

```
General::ivar: -0.918326 is not a valid variable. >>
```

```
General::stop:
```

```
Further output of General::ivar will be suppressed during this calculation. >>
```



Unfortunately, the "obvious" way to plot `D[Sin[x], x]` gave an error message about "not a valid variable." What went wrong? This is a *Mathematica* quirk: in doing a plot, it gives `x` a numerical value (say -0.99999), and then tries to compute `D[Sin[x], x]` with `x=-0.9999`. The derivative with respect to a *numerical value* is a nonsensical operation; you can only differentiate with respect to a *variable*. Hence, the error message about "not a valid variable."

So, how do you plot a derivative? Create a new function

```
df[x_] = D[Sin[x], x]
```

$\cos(x)$

and plot df.

```
Plot[df[x], {x, -1, 1}]
```

